

Excel. Устранение проблем с функцией ВПР

Это вторая глава книги Билла Джелена. [Всё о ВПР: от первого применения до экспертного уровня](#). Раз вы читаете эту главу, видимо, ваша ВПР возвращает ошибку #Н/Д. Этот симптом может быть вызван одной из нескольких проблем. Чтобы найти лекарство, посмотрите, что из следующего применимо к вашему случаю:

- Большинство функций ВПР работают, но есть несколько разрозненных #Н/Д.
- #Н/Д нет в верхней части ваших данных, но #Н/Д встречаются всё чаще, по мере того, как вы двигаетесь вниз по диапазону. Скорее всего, вы забыли вставить знаки доллара в формулу, чтобы зафиксировать ссылку на таблицу просмотра.
- Все ВПР возвращают #Н/Д.
- ВПР работают для текстовых ячеек, но не для числовых.

Разрозненные #Н/Д. Проблема: отдельные ВПР выдают ошибки #Н/Д (рис. 2.1).

	A	D	E	F
1	Наименование	Описание		Наименование
62	Cross50-5	14K Gold Onyx Cross with White Cubic Zirconia Stone		
63	CR50-1	14K Gold Onyx Cross		
64	CR50-3	14K Gold Cross with Onyx		
65	BG33-3	14K Gold Bangle Bracelet with Vine Design		
66	BG33-8	14K Gold Bangle Bracelet with Star Design		
67	BG33-9	#Н/Д		
68	CR50-6	14K Gold Onyx Cross with White Cubic Zirconia Stone		
69	ER46-22	14K Gold Hoop Earrings		
70	CR33-2	14K Gold Bangle Bracelet with Vine Design		

Рис. 2.1. BG33-9 – это новый товар, которого нет в таблице подстановки

Такие ситуации встречаются, когда искомое значение не найдено в первом столбце *Таблицы*. В нашем примере товар BG33-9 не найден в таблице постановки (в диапазоне F2:H29). Добавьте элемент BG33-9 в таблицу подстановки (рис. 2.2). Исправьте формулу ВПР, расширив диапазон таблицы =ВПР(A2;\$F\$2:\$H\$30;3;ЛОЖЬ).

	A	D	E	F	G	H
1	Наименование	Описание	Наименование	Цена	Описание	
26	RG75-3	14K Gold Ray Of Light Onyx Men's Ring	ER80-63	23,95	14K Gold Ruby Colored Cubic Zi	
27	CR50-6	14K Gold Onyx Cross with White Cubic Z	ER89-47	21,95	14K Gold Two-Tone Rose Earrin	
28	CR50-6	14K Gold Onyx Cross with White Cubic Z	P411A	49,95	14K Gold Lion Pendant	
29	ER46-28	14K Gold Earrings	P330	49,95	14K Gold Eagle Pendant	
30	BR26-3	18K Italian Gold Men's Bracelet	BG33-9	12,95	16K Gold Hoop/Tube Earrings	
31	ER41-4	14K Gold Swiss Cut Earrings				
32	BR15-3	14K Gold Onyx Men's Bracelet				
33	CR50-1	14K Gold Onyx Cross				

Рис. 2.2. Новый товар добавлен в таблицу подстановки

Чтобы при включении новых строк в таблицу подстановки не приходилось постоянно исправлять функцию ВПР, можно использовать один из следующих приемов:

- Вставьте новые ячейки в любом месте в середине таблицы подстановки. Например, выделите ячейки F28:H28, кликните правой кнопкой мыши, выберите опцию *Вставить...* и в открывшемся окне укажите *Ячейки со сдвигом вниз*.
- В качестве таблицы подстановки выберите не диапазон ячеек (\$F\$2:\$H\$29), а столбцы целиком – F:H. Теперь, вы можете добавлять элементы внизу таблицы подстановки без переписывания формул. Excel достаточно сообразителен, чтобы использовать только непустые ячейки при расчетах.
- Выберите любую ячейку в таблице подстановки и нажмите сочетание клавиш Ctrl+T (Т английское). Это превращает диапазон в таблицу. При вводе новых значений в пустые строки ниже таблицы, она автоматически расширится. Теперь ваша ВПР будет ссылаться не на диапазон, а на ТАБЛИЦУ (рис. 2.3).

ВПР		=ВПР(A2;Таблица4[#Все];3;ЛОЖЬ)			
A	D	E	F	G	H
1	Наименование	Описание	Наименование	Цена	Описание
2	BG33-9	=ВПР(A2;Таблица4[#Все];3;ЛОЖЬ)	BG33-3	39,95	14K Gold Bangle Bracelet with V
3	W25-6	18K Italian Gold Women's Watch	CR50-3	14,95	14K Gold Cross with Onyx
4	W25-6	18K Italian Gold Women's Watch	RG75-3	99,95	14K Gold Ray Of Light Onyx Men
5	W25-6	18K Italian Gold Women's Watch	RG78-25	99,95	14K Gold Ballerina Ring w/ Blue
6	W25-6	18K Italian Gold Women's Watch	W25-6	79,95	18K Italian Gold Women's Watch
7	BR26-3	18K Italian Gold Men's Bracelet	BR26-3	44,95	18K Italian Gold Men's Bracelet
8	BR26-3	18K Italian Gold Men's Bracelet	BR15-3	12,95	14K Gold Onyx Men's Bracelet
9	BR26-3	18K Italian Gold Men's Bracelet	BG33-8	12,95	14K Gold Bangle Bracelet with St
10	BR26-3	18K Italian Gold Men's Bracelet	BG33-17	12,95	14K Gold Bangle Bracelet with St
11	BR26-3	18K Italian Gold Men's Bracelet	CR 50-4	12,95	14K Gold Onyx Cross
12	BR26-3	18K Italian Gold Men's Bracelet	CR50-2	12,95	14K Gold Onyx Cross
13	BR26-3	18K Italian Gold Men's Bracelet	CR50-1	12,95	14K Gold Onyx Cross
14	ER41-4	14K Gold Swiss Cut Earrings	Cross50-5	12,95	14K Gold Onyx Cross with Whit
15	ER41-4	14K Gold Swiss Cut Earrings	CR50-6	12,95	14K Gold Onyx Cross with Whit
16	ER41-4	14K Gold Swiss Cut Earrings	ER41-4	22,95	14K Gold Swiss Cut Earrings
17	ER41-4	14K Gold Swiss Cut Earrings	ER46-14	22,95	14K Gold Fish Hoop Earrings
18	ER41-4	14K Gold Swiss Cut Earrings	ER46-7	20,95	14K Gold Hollow Earrings
19	RG75-3	14K Gold Ray Of Light Onyx Men's Ring	ER46-22	24,95	14K Gold Hoop Earrings
20	RG75-3	14K Gold Ray Of Light Onyx Men's Ring	ER46-28	22,95	14K Gold Earrings
21	RG75-3	14K Gold Ray Of Light Onyx Men's Ring	ER46-29	24,95	14K Gold Hoop Earrings
22	RG75-3	14K Gold Ray Of Light Onyx Men's Ring	ER 46-33	21,95	14K Gold Hollow Earrings
23	RG75-3	14K Gold Ray Of Light Onyx Men's Ring	ER49-20	22,95	14K Gold Hoop/Tube Earrings
24	RG75-3	14K Gold Ray Of Light Onyx Men's Ring	ER49-21	24,95	14K Gold Hoop/Tube Earrings
25	RG75-3	14K Gold Ray Of Light Onyx Men's Ring	ER49-22	20,95	14K Gold Tube/Hoop Earrings
26	RG75-3	14K Gold Ray Of Light Onyx Men's Ring	ER80-63	23,95	14K Gold Ruby Colored Cubic Zi
27	BR15-3	14K Gold Onyx Men's Bracelet	ER89-47	21,95	14K Gold Two-Tone Rose Earring
28	BR15-3	14K Gold Onyx Men's Bracelet	P411A	49,95	14K Gold Lion Pendant
29	Cross50-5	14K Gold Onyx Cross with White Cubic	P330	49,95	14K Gold Eagle Pendant
30	Cross50-5	14K Gold Onyx Cross with White Cubic	BG33-9	12,95	16K Gold Hoop/Tube Earrings
31	Cross50-5	14K Gold Onyx Cross with White Cubic Zirconia Stones			

Рис. 2.3. Превратите диапазон таблицы подстановки в инструмент Excel ТАБЛИЦА; это позволит автоматически расширять его при добавлении новых строк

Чтобы не пропустить ошибки #Н/Д, отсортируйте данные по столбцу *Описание* по убыванию (рис. 2.4). Все ошибки соберутся сверху. Для этого выделите столбец D, перейдите на вкладку *ДАННЫЕ* и в области *Сортировка и фильтры* нажмите кнопку сортировки. В открывшемся окне выберите *Автоматически расширить выделенный диапазон* и кликните *Сортировать*.




Рис. 2.4. Сортировка по убыванию разместит все записи с ошибкой #Н/Д сверху диапазона

Использование функций ЕСЛИОШИБКА, ЕСНД и ЕНД для исключения значений с #Н/Д. Проблема: в отличие от предыдущего примера, элементы с #Н/Д не приводят к потере данных и не требуют

дополнения таблицы подстановки. Если не найдено значение, желательно иметь пустую ячейку или ноль.

Существует несколько способов решить эту проблему. Если вы используете Excel 2007 или более позднюю версию, воспользуйтесь функциями ЕСЛИОШИБКА() или ЕСНД(). Если у вас более ранняя версия Excel (или требуется совместимость с более ранней версией) подойдет связка функций ЕСЛИ() и ЕНД().

Если исходная формула =ВПР(А67;F\$2:\$H\$29;3;ЛОЖЬ), то замените ее на =ЕСЛИОШИБКА(ВПР(А67;F\$2:\$H\$29;3;ЛОЖЬ);0), или на =ЕСНД(ВПР(А67;F\$2:\$H\$29;3;ЛОЖЬ);0).

Функция ЕСЛИОШИБКА() появилась в Excel 2007, что стало значительным улучшением по сравнению с предыдущими методами. С ЕСЛИОШИБКА() Excel сначала вычисляет формулу в первом аргументе функции. В 99% случаев возвращается правильный ответ, и работа функции останавливается. В тех редких случаях, когда возвращается ошибка, Excel переходит к обработке второго аргумента функции ЕСЛИОШИБКА(). В нашем примере, просто возвращает ноль.

Если вы предпочитаете в случае ошибки получить пустую ячейку, используйте формулу =ЕСЛИОШИБКА(ВПР(А67;F\$2:\$H\$29;3;ЛОЖЬ);""), но помните, что ячейка не будет по-настоящему пустой. Она будет содержать пустую текстовую строку, что, например, выдаст ошибку при попытке суммировать ее с другими числовыми значениями.

В более ранних версиях Excel 2007 используйте формулу: =ЕСЛИ(ЕНД(ВПР(А67;F\$2:\$H\$29;3;ЛОЖЬ));0;ВПР(А67;F\$2:\$H\$29;3;ЛОЖЬ)).

Это ужасная формула, чтобы вычислить ее, нужно вдвое больше времени. Для каждой ячейки Excel сначала вычислит ВПР. Если результат – ошибка #Н/Д, то подставится ноль. Если результат – не ошибка, Excel повторно вычислит ВПР. С тысячами ячеек, содержащими ВПР задержка по времени может быть заметной.

Удалите начальные и конечные пробелы. Проблема: ни одна из моих формул ВПР работает. Я четко вижу, что есть совпадение в таблице подстановки, но Excel не находит эти значения.

Когда вы сталкиваетесь с такой проблемой, скорее всего, почти все ячейки содержат #Н/Д, а не некоторые, как в предыдущем разделе. Проблема может заключаться в том, что, либо элементы в столбце А, либо в столбце F имеют пробелы. Это может произойти, если вы импортировали данные в Excel из 1С, SAP и т.п. Чтобы диагностировать эту проблему, выделите ячейку А2 и нажмите клавишу F2, чтобы перевести ячейку в режим редактирования. Мигающий курсор вставки появится в конце ячейки (рис. 2.5). Если курсор вставки появляется сразу после последнего символа, всё в порядке. Если, как на рис. 2.5, значит присутствуют лишние пробелы. Аналогичную диагностику можно провести и для ячеек в столбце F.

	A	D	E	F	G
1	Наименование	Описание			
2	BG33-8	#Н/Д		Наименование	Описание
3	Cross50-5	#Н/Д		BG33-3	14K Gold Bangl
4	RG78-25	#Н/Д		BG33-8	14K Gold Bangl
5	BG33-8	#Н/Д		CR50-3	14K Gold Cross
6	BG33-8	#Н/Д		RG75-3	14K Gold RAY
7	FR46-7	#Н/Д		RG78-25	14K Gold Baller

Рис. 2.5. Ни одна из ВПР не работает, хотя совпадения наблюдаются

Воспользуйтесь функцией СЖПРОБЕЛЫ() для удаления начальных и конечных пробелов. Если есть пробелы между словами, функция изменит несколько пробелов подряд на один. Таким образом, формулу =ВПР(А3;F\$3:\$G\$30;2;ЛОЖЬ) замените на =ВПР(СЖПРОБЕЛЫ(А3);F\$3:\$G\$30;2;ЛОЖЬ).

Если лишние пробелы отображаются в таблице подстановки, вы можете воспользоваться формулой массива. Наберите с клавиатуры =ВПР(А3;СЖПРОБЕЛЫ(F\$3:\$G\$30);2;ЛОЖЬ). Но не жмите Enter после редактирования. Вместо этого, нажмите одновременно Ctrl + Shift + Enter (подробнее см. [Майкл Гирвин. Ctrl+Shift+Enter. Освоение формул массива в Excel](#)). Ваша формула будет выглядеть так {=ВПР(А3;СЖПРОБЕЛЫ(F\$3:\$G\$30);2;ЛОЖЬ)}. Появятся фигурные скобки.

Подводный камень: формулы массива работают медленнее обычных формул. Это не проблема, если вы хотите произвести впечатление на коллег. Но в реальной жизни, лучше временно создать столбец,

например, H, в который ввести формулу =СЖПРОБЕЛЫ(F3). Затем скопировать его и вставить как значения в столбец F.

Числа и текст не распознаются как одинаковые. Проблема: часть значений ВПР работают корректно, а часть – возвращают ошибку (рис. 2.6).

	A	B	C	D	E	F	G
1	Дата	Наиме	Количе	Цена		Наименов	Цена
2	05.01.2014	1120TS	150	2195		610	275
3	19.01.2014	610	250	#Н/Д		610x	325
4	19.01.2014	1120	250	2095		700	500
5	06.02.2014	710x	200	375		710x	375
6	18.02.2014	610x	50	325		750	595
7	23.02.2014	750ES	250	695		750ES	695
8	04.03.2014	1120	250	#Н/Д		860	795
9	08.03.2014	610x	200	325		960	1495
10	11.03.2014	1120	100	2095		960LR	1695
11	16.03.2014	610	200	275		1120	2095
12	21.03.2014	960	250	#Н/Д		1120TS	2195
13	21.03.2014	960	100	1495			
14	10.04.2014	960	250	765			

Рис. 2.6. Часть значений ВПР возвращают ошибку

Вероятно, некоторые ячейки содержат числовые значения, а некоторые – числа, сохраненные как текст. На приведенном выше рисунке вы можете увидеть крошечный зеленый треугольник в левом верхнем углу ячеек B8, B11, B13. Если вы наведете курсор на ячейку, содержащую треугольник, слева появится символ с восклицательным знаком. Если раскрыть список, вы увидите, сообщение *Число, сохраненное как текст* (рис. 2.7).




Рис. 2.7. Сообщение: *Число, сохраненное как текст*

То, что числа хранятся в виде текста не обязательно плохо. Например, если это – модели продукта. Проблема в другом: формат данных в столбцах B и F – разный. Когда значение в B3 хранится в виде числа, а в F2 – в виде текста (см. рис. 2.6), ВПР не находит соответствия. Существуют функции, которые позволяют изменять формат данных. Однако, прежде чем их применить, нужно определить, в каком месте данные хранятся в виде чисел, а в каком – в виде текста. Для этого на свободном месте листа введите формулу =ЕЧИСЛО(B2) и =ЕЧИСЛО(F3). Значение ИСТИНА соответствует числу, ЛОЖЬ – тексту.

Если в столбце B хранится текст, а в столбце F – числа, используйте функцию ЗНАЧЕН, которая позволяет изменить текст на число. К сожалению, она выдаст ошибку, если наименование продукта буквенно-цифровое, например, 1120TS. Поэтому, нужно использовать конструкцию ЕСЛИОШИБКА вместе со ЗНАЧЕН (рис. 2.8).

D2 : fx =ВПР(ЕСЛИОШИБКА(ЗНАЧЕН(B2);B2);\$F\$2:\$G\$12;2;ЛОЖЬ)

	A	B	C	D	E	F	G	H	I	J
1	Дата	Наименов	Количество	Цена		Наименов	Цена			
2	05.01.2014	1120TS	150	2195		610	275			
3	19.01.2014	610	250	275		610x	325			
4	19.01.2014	1120	250	2095		700	500			
5	06.02.2014	710x	200	375		710x	375			
6			50	325		750	595			
7			250	695		750ES	695			
8			250	#Н/Д		860	795			
9			200	325		960	1495			
10	11.03.2014	1120	100	2095		960LR	1695			

Diagram: An orange arrow points from the word 'Текст' (Text) to cell B2. A blue arrow points from the word 'Число' (Number) to cell F2.

Рис. 2.8. Преобразования текстовых значений в числовые

Если ситуация обратная, то можно использовать функцию ТЕКСТ. Буквенно-цифровые значения при этом не изменятся (рис. 2.9). Второй аргумент функции ТЕКСТ ("0") задает формат числа, на основании которого произойдет конвертация в текст (подробнее см. [Пользовательский формат числа в Excel](#)).

1	Дата	Наименов	Количество	Цена	Наименов	Цена
2	05.01.2014	1120TS	150	2195	610	275
3	19.01.2014	610	250	275	610	325
4	19.01.2014	1120	250	2095	610	500
5	06.02.2014	710x	200			
6						
7						
8						
9					960	1495
10	11.03.2014	1120	100	2095	960LR	1695
11	16.03.2014	610	200	275	1120	2095

Рис. 2.9. Преобразования числовых значений в текстовые

Что делать, если проблема не укладывается ни в один из шаблонов? Т.е., когда, и столбец В, и столбец F, содержат комбинацию чисел и текста (рис. 2.10).

1	Дата	Наименов	Количество	Цена	Наименов	Цена
2	05.01.2014	1120TS	150	2195	610	275
3	19.01.2014	610	250	#Н/Д	610x	325
4	19.01.2014	1120	250	2095	700	
5	06.02.2014	710x	200	375	710x	
6	18.02.2014	610x	50	325	750	
7	23.02.2014	750ES	250	695	750ES	695
8	04.03.2014	1120	250	#Н/Д	860	795
9	08.03.2014	610x	200	325	960	
10	11.03.2014	1120	100	2095	960LR	
11	16.03.2014	610	200	275	1120	
12	21.03.2014	960	250	#Н/Д	1120TS	2195
13	21.03.2014	960	100	1495		
14	10.04.2014	860	250	795		
15	11.04.2014	700	100	500		

Рис. 2.10. Произвольное сочетание числовых и текстовых форматов в столбцах В и F

Если это произойдет, вы можете конвертировать числа, представленные текстом, в обоих столбцах. Выполните следующее:

1. Выделите ячейку B2.
2. Нажмите Ctrl+Shift+Стрелка вниз. Выделится весь столбец В с данными.
3. Наберите Alt+Y, не отпуская Alt, продолжайте набирать П, и наконец Г (или кликните на кнопке *Готово*). Это позволит преобразовать текстовые значения во всем столбце в числа.
4. Повторите пункты 1–3 для столбца F.

Примечание: сочетание клавиш Alt+Y+П+Г – это быстрый способ выбрать все настройки по умолчанию в *Мастере распределения текста по столбцам* (вкладка *ДААННЫЕ*, область *Работа с данными*). Использование *Мастера* позволяет преобразовать все данные обратно в общий формат, что, как правило, решает проблему.

Альтернативные стратегии. Существуют и иные способы преобразования текста в числа. Самый простой способ собрать все текстовые ячейки рядом – отсортировать данные по столбцу В по убыванию. Ячейки с текстом попадут в начало списка. Если текстовых ячеек немного, их можно обработать вручную.

В Excel 2002 и более поздних версиях, вы можете преобразовать непрерывный диапазон ячеек из текста в числа. Чтобы сделать это, вы используйте знак ошибки (восклицательный знак). Раскройте список и выберите *Преобразовать в число* (см. рис. 2.7). Этот метод работает только если верхняя ячейка вашего диапазона содержит число, сохраненное как текст.

Контроль дубликатов при использовании ВПР. Проблема: у меня есть первоначальная таблица, и данные о продажах за 19.06.14. Я использовал функцию ВПР, чтобы извлечь продажи из списка, и

отразить их в первоначальной таблице. Затем я получил файл с продажами за 20.06.14. Я воспользовался функцией ПОИСКПОЗ и нашел одного нового клиента в списке продаж 20.06.14 – Sun Life Fincl.

Sales for 6/19/2014		
Customer	Revenue	There?
Sun Life Fincl.	3541	#N/A
Molson, Inc	2944	12
Bell Canada	2834	8
Sun Life Financial	2815	20
Lucent	2528	2
Chevron	2362	7
Wal-Mart	2183	19

Рис. 2.11. Действительно ли это новый клиент?

Этого клиента новым не назовешь. Кто-то в департаменте закупок создал его вместо существующего клиента по имени Sun Life Financial. Чтобы исправить этот дубль, надо заменить в таблице Sun Life Fincl. на Sun Life Financial. Это устранил ошибку #Н/Д. Однако, после этого исправления у вас появилось две строки, которые соответствуют Sun Life Financial.

ВПР не способна обрабатывать ситуации, подобные описанной. При наличии двух строк в таблице подстановки ВПР вернет продажи из первой попавшейся. Вы получите \$3541, но не \$2815.

	A	B	C
25	Sun Life Financial	21,578	3,541
26	Ford	2,629	0

Рис. 2.12. ВПР возвращает первое найденное значение

Если у вас нет полной уверенности, что клиенты в таблице подстановки уникальны, вы не должны использовать ВПР. Вместо этого используйте функцию СУММЕСЛИ. На рис. 2.13 использована следующая формула: =СУММЕСЛИ(\$A\$2:\$A\$10;E2;\$C\$2:\$C\$10). Эта формула предписывает выполнять сканирование по столбцу *Клиент* (диапазон A2:A10), чтобы найти клиента, соответствующего значению в ячейке E2. Каждый раз, когда в Excel находит соответствие *Выручка* из диапазона C2:C10 добавляется к значению в ячейке F2.

The screenshot shows the Excel formula bar with the formula: `=СУММЕСЛИ(A2:A10;E2;C2:C10)`. Below the formula bar is a table with the following data:

	A	B	C	D	E	F	G
1	Клиент	Номер	Выручка		Клиент	Общая выручка	
2	AT&T	17061	153001		AT&T	=С\$10	
3	Chevron	17062	41002		Chevron	304016	
4	HP	17063	124003		HP	270009	
5	AT&T	17064	92004		P&G	164008	
6	Chevron	17065	98005				
7	HP	17066	146006				
8	AT&T	17067	137007				
9	P&G	17068	164008				
10	Chevron	17069	165009				
11							

Рис. 2.13. Используйте СУММЕСЛИ вместо ВПР для таблиц подстановки, содержащих дубли

Параметр *Искомое_значение* должен содержать не более 255 символов. Проблема: ни одна из моих ВПР не работает. Я получаю ошибку #ЗНАЧ!

Чтобы выяснить, сколько знаков в ячейке A2, перейдите в любую пустую ячейку и введите =ДЛСТР(A2). ВПР не будет работать, если символов >255. В определенных обстоятельствах вы могли бы усечь содержимое ячейки с помощью формулы =ЛЕВСИМВ(A2;255). Но, иногда это не удастся, если ключи не уникальны при усечении до 255 символов.

Таблица подстановки имеет неверную ориентацию. Проблема: таблица подстановки расположена по горизонтали (рис. 2.14). Можно ли использовать ВПР?

	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
1					Бонусный процент											
2	Месяц	Продажи	Бонус		Jan	Feb	Mar	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec
3	Jan	17753	0.015		1.5%	2.4%	3.0%	1.2%	1.9%	1.0%	2.1%	2.4%	2.2%	1.6%	2.5%	2.8%
4	Feb	16500														
5	Mar	26648														
6	Oct	10564														

Рис. 2.14. Таблица подстановки расположена по горизонтали

Стратегия: Буква “В” в ВПР обозначает вертикальный просмотр. В Excel также есть функция ГПР для горизонтальных таблиц подстановки. Если вы хотите почудачить, то могли бы использовать формулу =ГПР(В3,\$F\$2:\$Q\$3,2,ЛОЖЬ).

Альтернативная стратегия: вы, скорее всего, поступите также, как и большинство других пользователей. Скопируете диапазон F2:В3, выделите ячейку F5, выполните специальную вставку, задав параметр *Транспонировать*. Это предаст таблице подстановки вертикальное положение. Затем вы введете обычную формулу с ВПР. Заметьте, что, начиная с версии 2010 в Excel для транспонирования используется специальный значок:



Рис. 2.15. Значок Транспонировать